

Indicadores de Pobreza Monetaria con Stata

Luis Guillen Grados

Chief Data Science

lguillen@geoanalytics.pe


lguilleng@gmail.com




medium.com/@lguilleng



linkedin.com/in/datascientistlg



¿Cómo se mide la pobreza monetaria en el Perú



Una persona para poder desarrollar sus actividades cotidianas necesita ingerir 2.119 calorías por día. El costo promedio mensual, en el año 2014, de estas calorías (canasta alimentaria) es de S/.161.

Fuente: <https://archivo.elcomercio.pe/visor/1849816/1226484-cuales-son-metodos-usados-peru-medir-pobreza-noticia>

El indicador más utilizado para medir la pobreza monetaria es la incidencia de pobreza. Este indicador mide el porcentaje de personas cuyo gasto o ingreso per cápita está por debajo de la línea de pobreza en relación a la población total. A pesar de que este indicador tiene la ventaja de ser fácilmente comprensible, también tiene algunas limitaciones. Por ejemplo, no considera la intensidad de la pobreza, lo que implica que no refleja la gravedad de la situación de los pobres. En el Perú, el INEI utiliza este indicador en sus estadísticas oficiales de pobreza.

Indicadores de pobreza monetaria con Stata

Para estimar la pobreza en el Perú 2021, utilizaremos el archivo de datos sumaria-2021.dta de la Encuesta Nacional de Hogares realizada por el Instituto Nacional de Estadística e Informática (INEI). Este archivo y otros puede ser descargado del siguiente enlace: <https://proyectos.inei.gob.pe/microdatos/>

Para el cálculo de indicadores de pobreza monetaria, el INEI utiliza los propuesto por Foster-Greer-Thorbecke. Como se indicó en la lámina anterior, la expresión matemática que permite su cálculo es:

$$P_{\alpha}(Z,X) = \frac{1}{N} \sum \left(\frac{Z - X_i}{Z} \right)^{\alpha}$$

Donde N es el tamaño de la población, Z es el umbral de pobreza (línea de pobreza) y Xi, para el caso del Perú, es el gasto per cápita mensual..

```
. use sumaria-2021, clear
. describe gashog2d linpe linea
```

Variable name	Storage type	Display format	Value label	Variable label
gashog2d	double	%10.0g		gasto total bruto
linpe	float	%9.0g		línea de pobreza alimentaria
linea	float	%9.0g		línea de pobreza total

gasto total bruto ANUAL del hogar

valor de la canasta básico de consumo mensual por persona (alimentos)

valor de la canasta básico de consumo mensual por persona (alimentos + no alimentos)

Las líneas de pobreza total y alimentaria, están por mes y por persona, por lo que el indicador de bienestar (gasto) debe encontrarse en las mismas dimensiones. Entonces, necesitamos crear la variable “gasto per cápita mensual”

```
. g gpcm=gashog2d/12/mieperho
. label var gpcm "gasto per cápita mensual"
. sum gpcm
```

Variable	Obs	Mean	Std. dev.	Min	Max
gpcm	34,245	746.436	641.1798	45.02584	17060.33

```
. sum gpcm, d
```

gasto per cápita mensual				
Percentiles		Smallest		
1%	131.425	45.02584		
5%	196.6769	46.66385		
10%	248.6921	53.2571	Obs	34,245
25%	369.9207	54.43471	Sum of wgt.	34,245
50%	575.6034		Mean	746.436
		Largest	Std. dev.	641.1798
75%	910.5422	11153.28		
90%	1397.923	12494.81	Variance	411111.5
95%	1830.475	13392.22	Skewness	4.370296
99%	3245.301	17060.33	Kurtosis	47.05737

- en la muestra, ¿cuántas veces mas gasta una persona del 1% de hogares mas ricos respecto al 1% de hogares mas pobres?
- Si la línea de pobreza fuese 570 soles, ¿qué porcentaje aproximado de hogares pobres habría en la muestra?
- en la muestra, ¿cuál es el gasto diario por persona en el 1% de hogares mas pobres?

Cálculo de los indicadores de clase Pα (Foster-Greer-Thorbecke)

Pobreza Total

```
/* CALCULO DE LOS INDICADORES P0, P1 y P2 POR CADA HOGAR EN LA MUESTRA
   POBREZA MONETARIA TOTAL ---> Línea de pobreza total (Z) = linea */

/* Si Z-Xi <0, entonces indicador=0 */
gen indicador=0
replace indicador=1 if gpcm<linea /* identifica a los pobres */

/* Calculo de los indicadores FGT por hogar
   alpha=0 (INCIDENCIA),alpha=1 (BRECHA),alpha=2 (SEVERIDAD) */

gen p0=indicador*((linea-gpcm)/linea)^0
gen p1=indicador*((linea-gpcm)/linea)^1
gen p2=indicador*((linea-gpcm)/linea)^2

/* Estimación de los indicadores P0, P1 y P2 expandido al total de la población */

g facpob=factor07*mieperho /* Necesitamos factor de expansión de personas */
svyset conglome [pw=facpob], strata(estrato) /* Estructura compleja de la muestra */
svy: mean p0 p1 p2 /* Estimamos P0, P1 y P2 para el total de personas Perú 2021 */
```

Number of strata =	8	Number of obs =	34,245
Number of PSUs =	5,359	Population size =	33,271,455
		Design df =	5,351

		Linearized		
		Mean	std. err.	[95% conf. interval]
p0		.2587305	.005185	.2485658 .2688951
p1		.0644302	.0016979	.0611016 .0677588
p2		.0234309	.0008032	.0218563 .0250054

Sus CV son:

```
. estat cv
```

		Linearized		
		Mean	std. err.	CV (%)
p0		.2587305	.005185	2.00401
p1		.0644302	.0016979	2.63528
p2		.0234309	.0008032	3.42782

Pobreza Extrema

```
/* CALCULO DE LOS INDICADORES P0, P1 y P2 POR CADA HOGAR EN LA MUESTRA
   POBREZA MONETARIA EXTREMA ---> Línea de pobreza alimentaria (Z) = linpe */

/* Si Z-Xi <0, entonces indicador=0 */
gen indicador_e=0
replace indicador_e=1 if gpcm<linpe /* identifica a los pobres extremos */

/* Calculo de los indicadores FGT por hogar
   alpha=0 (INCIDENCIA), alpha=1 (BRECHA), alpha=2 (SEVERIDAD) */

gen p0_e=indicador_e*((linpe-gpcm)/linpe)^0
gen p1_e=indicador_e*((linpe-gpcm)/linpe)^1
gen p2_e=indicador_e*((linpe-gpcm)/linpe)^2

/* Estimación de los indicadores P0, P1 y P2 expandido al total de la población */

svy: mean p0_e p1_e p2_e /* EstimamosP0, P1 y P2 para el total de personas Perú 2021 */
```

Number of strata =	8	Number of obs =	34,245
Number of PSUs =	5,359	Population size =	33,271,455
		Design df =	5,351

		Linearized		
		Mean	std. err.	[95% conf. interval]
p0_e		.0412305	.0020852	.0371427 .0453183
p1_e		.0074559	.0004563	.0065614 .0083504
p2_e		.0022166	.0001727	.001878 .0025552

Sus CV son:

```
. estat cv
```

		Linearized		
		Mean	std. err.	CV (%)
p0_e		.0412305	.0020852	5.05739
p1_e		.0074559	.0004563	6.11989
p2_e		.0022166	.0001727	7.79182

Pobreza Total por Departamento

Según la Ficha Técnica de la ENAHO 2021, la muestra es del tipo probabilística, de áreas, estratificada, multietápica e independiente en cada departamento de estudio. Esto indica que, la selección de la muestra en un departamento no se encuentra relacionada con la selección de la muestra de otro departamento, por lo que la variabilidad de los estimadores en un departamento, solo dependen de las unidades primarias de muestreo seleccionadas en el departamento en cuestión. Entonces, la estimación de los errores de muestreo deben ser independientes para cada departamento. Debemos utilizar el enfoque condicional (Cochran, W. G. 1977. Sampling Techniques. 3rd ed. New York: Wiley).

Antes de estimar los indicadores FGT por departamento, veamos las diferencias entre la estimación del indicador P0 para el departamento Madre de Dios, con los enfoques condicional e incondicional

```
g md=substr(ubigeo,1,2)=="17"
svy: mean p0 if md==1 /* enfoque condicional */
svy, subpop(md): mean p0 /* enfoque no condicional */
```

	Enfoque	
	Condicional	Incondicional
# of strata	5	5
# os PSUs	101	2 872
# of obs	621	19 539
Design df	96	2 867
p0 (incidencia)	0.0770557	0.0770557
std. err.	0.0342895	0.0342364
cv	44.5%	44.4%



El enfoque condicional restringe la estimación de la varianza del estimador a solo la muestra que pertenece al departamento de Madre de Dios. En este escenario el # de PSUs es fijo (101 psu seleccionadas en Madre de Dios).

Para el cálculo de la varianza del estimador, el enfoque incondicional considera a todas las PSUs del diseño de muestreo (2,872). En este enfoque, el # de PSUs se conoce luego de aplicada la muestra.

Ahora, considerando el enfoque condicional (el # de PSUs no es aleatorio), calculamos la estimación de la incidencia de pobreza (p0) por departamento

```
/* Crear variable departamento */

g dep=substr(ubigeo,1,2)
destring dep, replace
label define dep ///
1 "Amazonas" ///
2 "Ancash" ///
3 "Apurímac" ///
4 "Arequipa" ///
5 "Ayacucho" ///
6 "Cajamarca" ///
7 "Prov. Const. del Callao" ///
8 "Cusco" ///
9 "Huancaavelica" ///
10 "Huánuco" ///
11 "Ica" ///
12 "Junín" ///
13 "La Libertad" ///
14 "Lambayeque" ///
15 "Lima" ///
16 "Loreto" ///
17 "Madre de Dios" ///
18 "Moquegua" ///
19 "Pasco" ///
20 "Piura" ///
21 "Puno" ///
22 "San Martín" ///
23 "Tacna" ///
24 "Tumbes" ///
25 "Ucayali", replace

label value dep dep
label dep "departamento"

/* Calculo de la estimación de P0 por departamento */

forvalues i = 1/25 {
    di _n "departamento: `i' `': label dep `i'"
    svy: mean p0 if dep=="`i'"
}
```

```

/* Cuadro con la estimación de P0 por departamento */

collect clear

forvalues i = 1/25 {
    quietly: collect _r_b _r_se _r_ci: svy: mean p0 if dep==`i'
}

collect style cell result[_r_b _r_se _r_ci], nformat(%8.5f)
collect label levels result _r_b "Incidencia", modify
collect label levels cmdset ///
1 "Amazonas" ///
2 "Ancash" ///
3 "Apurímac" ///
4 "Arequipa" ///
5 "Ayacucho" ///
6 "Cajamarca" ///
7 "Prov. Const. del Callao" ///
8 "Cusco" ///
9 "Huancaavelica" ///
10 "Huánuco" ///
11 "Ica" ///
12 "Junín" ///
13 "La Libertad" ///
14 "Lambayeque" ///
15 "Lima" ///
16 "Loreto" ///
17 "Madre de Dios" ///
18 "Moquegua" ///
19 "Pasco" ///
20 "Piura" ///
21 "Puno" ///
22 "San Martín" ///
23 "Tacna" ///
24 "Tumbes" ///
25 "Ucayali", modify

collect layout (cmdset) (result)

```

singleunit(certainty)

```
. collect layout (cmdset) (result)
```

```

Collection: default
  Rows: cmdset
  Columns: result
  Table 1: 25 x 3

```

	Incidencia	Std. error	95% CI	
Amazonas	0.30096	0.02238	0.25679	0.34513
Ancash	0.21862	0.01734	0.18442	0.25282
Apurímac	0.28298	0.02517	0.23318	0.33278
Arequipa	0.12043	0.01435	0.09216	0.14869
Ayacucho	0.36019	0.02166	0.31743	0.40294
Cajamarca	0.39734	0.02122	0.35550	0.43918
Prov. Const. del Callao	0.28525	0.02292	0.23999	0.33051
Cusco	0.22005	0.02100	0.17862	0.26149
Huancaavelica	0.41206	0.02791	0.35687	0.46725
Huánuco	0.35489	0.02491	0.30573	0.40404
Ica	0.06573	0.01204	0.04202	0.08945
Junín	0.26364	0.01926	0.22569	0.30160
La Libertad	0.26763	0.01747	0.23321	0.30205
Lambayeque	0.13960	0.00000		
Lima	0.24619	0.01203	0.22257	0.26981
Loreto	0.34607	0.02056	0.30555	0.38659
Madre de Dios	0.07706	0.03429	0.00899	0.14512
Moquegua	0.10174	0.01418	0.07374	0.12973
Pasco	0.42115	0.03575	0.35041	0.49189
Piura	0.25306	0.01716	0.21926	0.28686
Puno	0.42597	0.00000		
San Martín	0.20314	0.01977	0.16415	0.24213
Tacna	0.21762	0.02610	0.16616	0.26907
Tumbes	0.19335	0.02276	0.14833	0.23837
Ucayali	0.17023	0.02195	0.12691	0.21356

Si necesitamos exportar la tabla anterior a formato MS-Excel, el comando es:

```
collect export tabla.xls
```

Incidencia de pobreza, según departamento. 2021

	(porcentaje)				
	Incidencia	Std. Error	95% CI		CV
Amazonas	30.096	2.238	25.679	34.513	7.44
Ancash	21.862	1.734	18.442	25.282	7.93
Apurímac	28.298	2.517	23.318	33.278	8.89
Arequipa	12.043	1.435	9.216	14.869	11.92
Ayacucho	36.019	2.166	31.743	40.294	6.01
Cajamarca	39.734	2.122	35.550	43.918	5.34
Prov. Const. del Callao	28.525	2.292	23.999	33.051	8.04
Cusco	22.005	2.100	17.862	26.149	9.54
Huancavelica	41.206	2.791	35.687	46.725	6.77
Huánuco	35.489	2.491	30.573	40.404	7.02
Ica	6.573	1.204	4.202	8.945	18.32
Junín	26.364	1.926	22.569	30.160	7.31
La Libertad	26.763	1.747	23.321	30.205	6.53
Lambayeque	13.960	-	-	-	-
Lima	24.619	1.203	22.257	26.981	4.89
Loreto	34.607	2.056	30.555	38.659	5.94
Madre de Dios	7.706	3.429	0.899	14.512	44.50
Moquegua	10.174	1.418	7.374	12.973	13.93
Pasco	42.115	3.575	35.041	49.189	8.49
Piura	25.306	1.716	21.926	28.686	6.78
Puno	42.597	-	-	-	-
San Martín	20.314	1.977	16.415	24.213	9.73
Tacna	21.762	2.610	16.616	26.907	12.00
Tumbes	19.335	2.276	14.833	23.837	11.77
Ucayali	17.023	2.195	12.691	21.356	12.89

Nota: El cálculo del error estándar para los departamento de Lambayeque y Puno no ha sido posible, debido a que existen estratos con un solo conglomerado.

Fuente: Encuesta Nacional de Hogares - 2021. INEI

Elaboración propia

¿Por qué el cuadro anterior no presenta errores de muestreo para los departamento de Lambayeque y Puno?

```
. svydes if dep==14
```

<omitido>

Stratum	# units	# obs	Number of obs per unit		
			Min	Mean	Max
1	62	425	4	6.9	11
2	27	172	4	6.4	9
3	8	49	6	6.1	7
4	27	165	4	6.1	9
5	51	306	5	6.0	7
6	1*	8	8	8.0	8
7	36	285	5	7.9	10
8	3	24	8	8.0	8
8	215	1,434	4	6.7	11

```
. svydes if dep==21
```

<omitido>

Stratum	# units	# obs	Number of obs per unit		
			Min	Mean	Max
2	58	283	2	4.9	8
4	1*	6	6	6.0	6
5	36	183	2	5.1	6
6	9	68	5	7.6	8
7	65	481	4	7.4	8
8	16	122	5	7.6	8
6	185	1,143	2	6.2	8

IMPORTANTE: Si, para la estimación por departamento, se hubiera utilizado la opción `over()`, se mostrarían errores estándar para Lambayeque y Puno. Pero el cálculo no sería correcto, ya que cada la selección de la muestra es independiente en cada departamento.

¿Qué nos dice el INEI sobre los departamentos Ica y Madre de Dios, que presentan los mas altos CV para la incidencia de pobreza?

2021	Grupo 1	Ayacucho, Cajamarca, Huancavelica, Huánuco, Loreto, Pasco, Puno	36,7	40,9
	Grupo 2	Amazonas, Apurímac, Junín, La Libertad, Lima ³ , Lima Metropolitana ² , Piura, Provincia Constitucional del Callao	24,0	27,1
	Grupo 3	Áncash, Cusco, San Martín, Tacna, Tumbes, Ucayali	19,0	22,8
	Grupo 4	Arequipa, Lambayeque, Moquegua	11,0	14,6
	Grupo 5	Ica, Madre de Dios	4,5	9,0

Fuente: Evolución de la Pobreza Monetaria 2010-2021, Informe Técnico. INEI

```
. svy: mean p0 if dep==11 | dep==17  
(running mean on estimation sample)
```

Survey: Mean estimation

```
Number of strata = 7          Number of obs = 2,163  
Number of PSUs   = 348       Population size = 986,232.66  
                               Design df      = 341
```

		Linearized		
	Mean	std. err.	[95% conf. interval]	
p0	.0675266	.011502	.0449027 .0901504	

¿Esta agrupación es válida, sabiendo que el CV de Madre de Dios es muy alto?

Dominancia estocástica

De acuerdo a los resultados obtenidos por departamento, se observa que, la incidencia de pobreza en Cajamarca es mayor a la de La Libertad. Además, sus intervalos de confianza no se traslapan, por lo que, la diferencia entre sus incidencias de pobrezas es estadísticamente significativa. Como se aprecia en el cuadro siguiente:

	Incidencia	Std. error	95% CI	
Cajamarca	0.39734	0.02122	0.35550	0.43918
La Libertad	0.26763	0.01747	0.23321	0.30205

La relación de orden entre la incidencia de pobreza de Cajamarca y La Libertad (39,7% > 26,8%) se sustentan en una línea de pobreza fijada para cada departamento. El siguiente cuadro muestra los valores de las líneas de pobreza total.

CAJAMARCA

```
. sum linea if dep==6
```

Variable	Obs	Mean	Std. dev.	Min	Max
linea	1,453	290.5304	28.34	265.4722	364.1123

LA LIBERTAD


```
. sum linea if dep==13
```

Variable	Obs	Mean	Std. dev.	Min	Max
linea	1,538	347.7573	40.41847	277.3391	374.7922

Interrogante:


¿Si las líneas tuvieran otro valor, la relación de orden se mantendría?

Para dar respuesta a la interrogante anterior, necesitamos realizar una comprobación de Dominancia Estocástica.



La condición de dominancia de primer orden indica que la pobreza es más baja (o más alta) en un dominio si su curva de incidencia de la pobreza se encuentra siempre por debajo de la curva de otro dominio.

Valor relativo = indicador de bienestar / línea de pobreza



Si las curvas se interceptan significará que el ranking de pobreza no es muy claro, ya que para un nivel de la línea de la pobreza la incidencia de un dominio respecto a la de otro dominio puede ser mayor y para otro nivel de la línea la incidencia puede ser menor.

```
/* DOMINANCIA ESTOCASTICA */

/* Creamos la variable que mide el número de veces que el indicador de bienestar se encuentra en la línea de pobreza
gpcm / línea, cuando esta relación es igual a 1, estará indicando el valor de la línea de pobreza. */


generate zr=gpcm/línea

/* Creamos las frecuencias relativas acumuladas de la población para Cajamarca y La Libertad */

cumul zr if dep==6 [aw=facpob], generate(pob_6)
cumul zr if dep==13 [aw=facpob], generate(pob_13)
label var pob_6 "Cajamarca"
label var pob_13 "La Libertad"

/* Graficamos */

line pob_6 pob_13 zr if zr<7, sort xline(0.5 1 1.5) xlab(0 0.5 1 1.5 2 4 6 8)
ytitle("Prop. acum. población") xtitle("gasto per cápita mensual / línea de pobreza")
text(0.5 3 "En el intervalo de una línea relativa de 0.5 y 1.5, \nla relación de mantiene", place(NE))
```



La brecha de pobreza monetaria en el Perú

```
. svyset conglome [pw=facpob], strata(estrato)
```

```
. svy: mean p1  
(running mean on estimation sample)
```

Survey: Mean estimation

```
Number of strata =      8      Number of obs   =    34,245  
Number of PSUs   = 5,359      Population size = 33,271,455  
Design df        =          5,351
```

		Linearized		
	Mean	std. err.	[95% conf. interval]	
p1	.0644302	.0016979	.0611016	.0677588

```
. estat cv
```

	Mean	Linearized	
		std. err.	CV (%)
p1	.0644302	.0016979	2.63528

En promedio, los pobres necesitan una transferencia monetaria de 6,4% de la línea de pobreza para acercarse a la línea

La estimación de la brecha presenta alta precisión (CV=2,6%)

¿cuál es la incidencia y brecha de pobreza monetaria en la población que reside en hogares con no mas de 4 miembros.

```
. svy, subpop(if mieperho<=4): mean p0 p1  
(running mean on estimation sample)
```

Survey: Mean estimation

```
Number of strata =      8      Number of obs   =    34,245  
Number of PSUs   = 5,359      Population size = 33,271,455  
Subpop. no. obs  =    26,486  
Subpop. size     = 19,953,927  
Design df        =          5,351
```

	Mean	Linearized		
		std. err.	[95% conf. interval]	
p0	.1704829	.0046741	.1613197	.1796461
p1	.0387636	.0013398	.036137	.0413903

Otra forma sería:

```
. g subpop=mieperho<=4  
. svy: mean p0 p1, over(subpop)  
(running mean on estimation sample)
```

Survey: Mean estimation

```
Number of strata =      8      Number of obs   =    34,245  
Number of PSUs   = 5,359      Population size = 33,271,455  
Design df        =          5,351
```

	Mean	Linearized		
		std. err.	[95% conf. interval]	
c.p0@subpop				
0	.3909537	.0092253	.3728683	.409039
1	.1704829	.0046741	.1613197	.1796461
c.p1@subpop				
0	.102887	.0032181	.0965783	.1091957
1	.0387636	.0013398	.036137	.0413903

¿Por qué no se utiliza IF?

Descomposición de los indicadores de clase Pα (Foster-Greer-Thorbecke)

- La descomposición de los indicadores de pobreza puede adoptarse cuando se quiere establecer el perfil de pobreza del país o cuando se busca saber cual es la contribución de diferentes subgrupos de la población al indicador nacional de la pobreza.
- Sin embargo, las técnicas de descomposición adquieren todo un significado cuando se trata de precisar la evolución en el tiempo de la pobreza.
- Las descomposiciones de la evolución de la pobreza permiten distinguir el impacto de cambios en la estructura de la población y cambios en la incidencia de la pobreza para cada uno de los subgrupos.

Sea una muestra de “n” hogares o individuos (i=1,2,...,n) con gastos y_i y ponderaciones w_i.

Sea f_i = w_i * N_i, donde

$$N = \sum_{i=1}^n w_i$$

Cuando las observaciones no son ponderadas entonces w_i = 1 y N = n

La línea de pobreza es z y la brecha de pobreza para el individuo i-ésimo es max(0, z-y_i)

Como se vio anteriormente, los índices de pobreza de la clase FGT están dados por:

$$FGT_{\alpha}(Z, X) = \frac{1}{N} \sum \left(\frac{Z - X_i}{Z} \right)^{\alpha}$$

$$FGT_{\alpha}(Z, X) = w_i I_i \sum \left(\frac{Z - X_i}{Z} \right)^{\alpha} \qquad I_i = \begin{cases} 1 & y_i < z \\ 0 & y_i \geq z \end{cases}$$

Suponiendo que se puede repartir de manera exhaustiva la población en K grupos mutuamente excluyentes (k = 1, ..., K), cada índice FGT puede descomponerse aditivamente en:

$$FGT_{\alpha} = \sum_{k=1}^k P_k FGT_{k\alpha}$$

Donde P_k = N_k/N es el número de personas en el grupo k dividido por el número total de personas (es decir, simplemente la proporción de personas en el grupo k). FGT_{kα} es el índice de pobreza para el subgrupo.

Para descomponer los indicadores FGT en Stata, podemos utilizar el comando povdeco

Total País

```
. povdeco gpcm [aw=facpob], varpl(linea)

Total number of observations = 34245
Weighted total no. of observations = 33271455
Number of observations poor = 6340
Weighted no. of obs poor = 8608338.9
Mean of gpcm amongst the poor = 276.992
Mean of poverty gaps (poverty line - gpcm) amongst the poor = 89.901

Foster-Greer-Thorbecke poverty indices, FGT(a)

-----+-----
All obs |          a=0          a=1          a=2
-----+-----
          | 0.25873    0.06443    0.02343
-----+-----

FGT(0): headcount ratio (proportion poor)
FGT(1): average normalised poverty gap
FGT(2): average squared normalised poverty gap
```

Descomponiendo la pobreza por Costa, Sierra, Selva y Lima Metropolitana

```
g region=dominio<=3
replace region=2 if dominio>=4 & dominio<=6
replace region=3 if dominio==7
replace region=4 if dominio==8
label define region 1 "Costa" 2 "Sierra" 3 "Selva" 4 "Lima Metropolitana"
label value region región

. povdeco gpcm [aw=facpob], varpl(linea) by(region)

Foster-Greer-Thorbecke poverty indices, FGT(a)
```

All obs	a=0	a=1	a=2
	0.25873	0.06443	0.02343

FGT(0): headcount ratio (proportion poor)
FGT(1): average normalised poverty gap
FGT(2): average squared normalised poverty gap

Proporción de población que pertenecen a cada subgrupo de análisis (región).

Medias aritméticas del gasto total per cápita para cada subgrupo de análisis (región).

Muestra las medias aritméticas de las brechas del gasto total per cápita con respecto a la línea de pobreza total, para cada subgrupo de análisis.

Decompositions by subgroup

Summary statistics for subgroup k = 1,...,K

region	Pop. share	Mean	Mean poor	Mean gap poor
Costa	0.23518	673.30979	295.39042	75.32449
Sierra	0.30992	495.05193	218.59396	80.94631
Selva	0.12607	542.49322	231.86801	81.05324
Lima Metropolitana	0.32884	859.36181	357.66487	112.20909

Medias aritméticas del gasto total per cápita de la población pobre, para cada subgrupo de análisis (región).

Subgroup FGT index estimates, FGT(a)

region	a=0	a=1	a=2
Costa	0.18278	0.03729	0.01146
Sierra	0.32499	0.08845	0.03419
Selva	0.26411	0.06925	0.02603
Lima Metropolitana	0.24854	0.05935	0.02086

Subgroup poverty 'share', $S_k = v_k.FGT_k(a) / FGT(a)$

region	a=0	a=1	a=2
Costa	0.16615	0.13610	0.11499
Sierra	0.38929	0.42548	0.45218
Selva	0.12869	0.13550	0.14005
Lima Metropolitana	0.31588	0.30292	0.29279

Subgroup poverty 'risk' = $FGT_k(a) / FGT(a) = S_k / v_k$

region	a=0	a=1	a=2
Costa	0.70646	0.57870	0.48893
Sierra	1.25609	1.37287	1.45903
Selva	1.02079	1.07486	1.11090
Lima Metropolitana	0.96061	0.92119	0.89037

Riesgos de pobreza para cada subgrupo de análisis (región). Es decir, nos dice en qué medida los pobres de un subgrupo están sub (o sobre) representados respecto a la parte que ellos representan en el conjunto de la población.